# Maintainer: Antonio Rojas <arojas@archlinux.org>
# Contributor: Giovanni Scafora <giovanni@archlinux.org>
# Contributor: Emanuele Rampichini <emanuele.rampichini@gmail.com>

pkgname=smb4k
pkgver=3.2.1
pkgrel=1
pkgdesc='A KDE program that browses samba shares'
arch=(loong64 x86_64)
url='https://smb4k.sourceforge.net/'
license=(GPL)
depends=(kparts kdnssd hicolor-icon-theme smbclient)
makedepends=(extra-cmake-modules kdoctools plasma-framework)
source=(https://downloads.sourceforge.net/project/$pkgname/$pkgver/$pkgname-$pkgver.tar.xz)
sha256sums=('a21211bd3c61a961037fe00a07bad2ee5e4ffe9b194739d8d46849a6ce023b48')

build() {
  cmake -B build -S $pkgname-$pkgver \
    -DBUILD_TESTING=OFF
  cmake --build build
}

package() {
  DESTDIR="$pkgdir" cmake --install build
}
